# How to solve a transcendal equation

I want to get the solution x of this equation

a (-(1/(1 + b c + a x)) - Log[b c + a x] + Log[1 + b c + a x]) - d = 0


in terms of a, b, c, d. Solve and FindRoot do not give any answer!

Clear[a, b, c, x, d, r]
eq1 = a (-(1/(1 + b c + a x)) - Log[b c + a x] + Log[1 + b c + a x]) - d == 0


To simplify, make common sub-expression substitution

eq2 = eq1 /. (b c + a x) -> z


Write it as

eq2 = (-(1/(1 + z)) - Log[z] + Log[1 + z]) == d/a


eq3 = eq2 /. (d/a) -> r


Now take the exponential

eq4 = Exp[#] & /@ eq3


eq5 = eq4 /. {(1 + z) -> k, z -> k - 1}


Now reduce comes to help. We can either use Reduce on Reals or not. For the general case

Reduce[eq5, k]


Using the last case for example:

eq6 = k == 1/(1 + ProductLog[C[1], -E^(-1 - r)])


Now start back substitution

 eq7 = eq6 /. {k -> 1 + z, r -> (d/a)}
eq8 = eq7 /. z -> (b c + a x)


Now solve for x

 Solve[eq8, x]
(*x -> (-1 - b c + 1/(1 + ProductLog[C[1], -E^(-1 - d/a)]))/a *)


But notice there is arbitrary parameters C[1] which is an integer. So there are infinite number of solutions. Also, the other Reduce results are still applicable in this final solution